
Introduction 

This application note describes a typical method for creating chorus effects on the DRE.  A 
chorus effect adds a slowly time-varying delay and pitch (less than 1Hz) to an original voice to 
simulate additional voices (a chorus). 
 
 

Algorithm 

The basic parameters of a chorus are the modulation depth, modulation rate and predelay.  The 
DRE's LFOs are used in sine mode to modulate the delay of the signal.  Maximum delay is 
determined by the amplitude of the LFO, or modulation depth.  As the LFO sine wave crosses 
zero, maximum change in delay versus time results in maximum change in pitch.  This rate of 
delay change is determined by both modulation depth and modulation rate.  The generated 
chorus is added back to the original dry signal with a certain amount of constant pre-delay to 
avoid objectionable phase cancellation artifacts. 

A chorus effect requires delay memory for each channel and at least one LFO set to sine mode.  
The output of the LFO can be thought of as a 13 bit integer part that specifies address offset 
(LFOA) and a 7 bit fractional part (LFOK) that is the coefficient used to interpolate between 
adjacent memory locations.  The interpolation is mandatory for smooth change in delay. 

The complete chorus effect requires two chorus commands.  The pair of commands calculates 
the average of two adjacent memory locations weighted by the fractional part of the LFO (LFOK).  
The first chorus command is passed the middle location of the delay line and the second is 
passed the adjacent-to-middle location.  The first command also latches in the current LFO 
value and complements the fractional portion (COMPK) for use in the second chorus command.  
The actual memory location read is the location passed plus an offset determined by the integer 
part of the LFO (LFOA). 
 
 chorus = value(address+LFOA)*LFOK + value(address+1 +LFOA)*(1-LFOK) 
 
If the LFO is set to its maximum amplitude (32767) it will add 4095 or subtract 4096 to the 
specified address, requiring a memory block size of 8192.  Smaller modulation amplitudes 
(typical) can get away with a smaller memory block. 

The implementation described mixes the dry signal with the chorus within the DRE, so no 
further mixing should be performed externally to evaluate the effect.  The mixing is done 
internally to give the programmer complete control over predelay in order to hear its effect.  In a 
finished implementation, the DRE is typically used only to generate the chorus, mixing is 
adjustable externally and predelay must take into account the different signal path delays of the 
dry and processed audio.  Conveniently the different signal path delays often provide a good 
amount pre-delay on their own. 

The implementation described is a Wide Stereo Chorus.  Left and right channels are processed 
independently with independent LFOs.  Starting with a stereo recording of a single voice, this 
will simulate two additional voices, one on each channel.  A Stereo Chorus would use one LFO 
for both channels and starting with a stereo recording of a single voice, would create one 
additional stereo voice. 

Tuning a Chorus Effect 

The amount of delay modulation is proportional to the modulation depth (LFO amplitude); 
however, the amount of pitch modulation is proportional to the product of modulation depth and 
modulation rate (LFO frequency).  So as you increase the LFO amplitude for a stronger effect, 
you may have to reduce the LFO frequency to avoid obvious warbling. 

The mix of the chorus to the dry signal results in a time-varying comb-filter cancellation effect.  
If the predelay is too short, the comb-filter cancellation artifacts will become audible and 
objectionable.  If the predelay is too large, the chorus will sound objectionably out of sync with 
the dry signal. 

Modulation rate, modulation depth and predelay of the chorus relative to dry signal all 
contribute to sound quality.  Beyond tuning these basic parameters, typical enhancements for 
more natural sounding chorus include adding a high-pass filter to the chorus to prevent low-
frequency beating against the dry signal, and adding a low-pass filter to the chorus to prevent 
strident high-frequency interference with the dry signal. 
 

Source Code 

; File:         AN320103.ASM 
; Description:  Wide Stereo Chorus Example 
; Authors:      Jeff Rothermel 
; Copyright 2001 Wavefront Semiconductor 
;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; ;;;;;;;;;;;;;;;;;;;;;;;;;;;; 
LFO0 SIN AMP=10000 FREQ=2 ; f = FREQ * 0.029Hz for Fs=48kHz 
LFO1 SIN AMP=10000 FREQ=3 
; 
MEM chorusmeml 8192  ; 8192 big enough for full AMP  LFO 
MEM chorusmemr 8192  ; right chorus memory 
;NOTE: memory locations are referenced by: 
; name Start of memory block 
; name' End of memory block 
; name" Middle of memory block 
;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; ;;;;;;;;;;;;;;;;;;;;;;;;;;;; 
RZP  ADCL  K=.5  ; Read left/2 into accumulator 
WZP  chorusmeml   ; Write acc to start left chorus mem 
RZPB  chorusmeml+400   ; Read delayed left to B reg  
; 
CHR0 RZP chorusmeml" COMPK LATCH ; Read middle of c horus memory 
CHR0 RAP chorusmeml"+1   ; Read middle+1 chorus mem ory 
; 
WBP  OUTL  K=.999  ; Write dry (B) + chorus (acc) t o OUTL 
;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; ;;;;;;;;;;;;;;;;;;;;;;;;;;;; 
RZP  ADCR  K=.5  ; Read right/2 into accumulator 
WZP  chorusmemr   ; Write acc to start right chorus  mem 
RZPB  chorusmemr+400   ; Read delayed right to B re g 
; 
CHR1 RZP chorusmemr" COMPK LATCH ; Read middle of c horus memory 
CHR1 RAP chorusmemr"+1   ; Read middle+1 chorus mem ory 
; 
WBP  OUTR  K=.999  ; Write dry (B) + chorus (acc) t o OUTR 
;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;; ;;;;;;;;;;;;;;;;;;;;;;;;;;;; 
RZP 0x00                          ;Add 16 extra rea ds for refresh 
RZP 0x40 
RZP 0x80 
RZP 0xc0 
RZP 0x100 
RZP 0x140 
RZP 0x180 
RZP 0x1c0 
RZP 0x200 
RZP 0x240 
RZP 0x280 
RZP 0x2c0 
RZP 0x300 
RZP 0x340 
RZP 0x380 
RZP 0x3c0 

NOTICE 

Wavefront Semiconductor reserves the right to make changes to their products 
or to discontinue any product or service without notice.  All products are sold 
subject to terms and conditions of sale supplied at the time of order 
acknowledgement.  Wavefront Semiconductor assumes no responsibility for the 
use of any circuits described herein, conveys no license under any patent or 
other right, and makes no representation that the circuits are free of patent 
infringement.  Information contained herein is only for illustration purposes and 
may vary depending upon a user’s specific application.  While the information in 
this publication has been carefully checked, no responsibility is assumed for 
inaccuracies. 

Wavefront Semiconductor products are not designed for use in applications 
which involve potential risks of death, personal injury, or severe property or 
environmental damage or life support applications where the failure or 
malfunction of the product can reasonably be expected to cause failure of the life 
support system or to significantly affect its safety or effectiveness. 

All trademarks and registered trademarks are property of their respective owners. 

